

```

// Un-Orte
//
// A Generative Sound Environment for the Studio Foyer
// of the Academy of the Arts Berlin Hanseatenweg
//
// (c) 2005 by Gerhard Eckel

ProxySpace.push(s);
b = Buffer.read(s, "/Users/eckel/Un-Orte.aif");

~gvol = -3.dbamp;
~ggvar = { LinExp.kr(LFDNoise3.kr(0.01), -1, 1, 0.01, 10) };
~grate = { LFDNoise3.kr(~ggvar) };
~gspeed = { LFDNoise3.kr(~ggvar) };
~gdurfac = { LFDNoise3.kr(~ggvar) };
~glevel = { LFDNoise3.kr(~ggvar, 0.5, 0.5) };
~gsync = { LFDNoise3.kr(~ggvar, 0.5, 0.5) };

~voices = { Array.fill(7, { arg i;
  var rate, pos, durfac, pan, speed, trigger, grains, level, sr = SampleRate.ir;
  rate = LinExp.kr(LFDNoise3.kr(~ggvar).balance(~grate, ~gsync), -1, 1, 1, 100);
  speed = LinExp.kr(LFDNoise3.kr(~ggvar).balance(~gspeed, ~gsync), -1, 1, 0.1, 10) / sr;
  durfac = LinExp.kr(LFDNoise3.kr(~ggvar).balance(~gdurfac, ~gsync), -1, 1, 0.1, 10);
  level = LFDNoise3.kr(~ggvar, 0.5, 0.5).balance(~glevel, ~gsync) ** 4;
  trigger = Impulse.ar(rate); pan = ToggleFF.ar(trigger) * 1.99 - 1;
  pos = Phasor.ar(Impulse.ar(speed), speed, 0, BufDur.kr(b.bufnum));
  grains = TGrains.ar(2, trigger, b.bufnum, 1, pos, durfac / rate, pan, 1, 1) / (durfac.sqrt);
  Out.ar(i * 2, OneZero.ar(grains * level * ~gvol, -0.4));
}).flat
};

```